

```
> restart:with(plots):
```

## S. Timoshenko "Analysis of bi-metal thermostats"

Setup: bi-metal strip with thickness  $h$ , uniformly heated by  $\Delta T$  consisting of 2 metals with coefficients of expansion  $\alpha_T$ , Young's moduli  $E$ , thicknesses  $thk$ , modified by additional widths

here:

Layer 1 is passive (0 % strain), material PLA

Layer 2 is active with variable % temperature strain, material LAYWOOD meta5

### EQUILIBRIUM:

of forces in layers:  $P_1 = P_2 (=P)$

of moments in layers:  $P \cdot h/2 = M_1 + M_2$

internal forces, no external forces

```
> P_1 := P;  
P_2 := P;  
EQU1 := M_1 + M_2 = P*h/2;
```

$$P_1 := P$$

$$P_2 := P$$

$$EQU1 := M_1 + M_2 = \frac{P h}{2} \quad (1)$$

### MATERIAL:

$M = EI \cdot \kappa = EI \cdot 1/r$

```
> MAT1 := M_1 = E_1*I_1/r;  
MAT2 := M_2 = E_2*I_2/r;
```

$$MAT1 := M_1 = \frac{E_1 I_1}{r}$$

$$MAT2 := M_2 = \frac{E_2 I_2}{r} \quad (2)$$

Combining EQU and MAT

```
> EQUMAT:=subs({MAT1,MAT2},EQU1);
```

$$EQUMAT := \frac{E_1 I_1}{r} + \frac{E_2 I_2}{r} = \frac{P h}{2} \quad (3)$$

### KINEMATICS:

on the bearing surfaces the total elongation/strain of both metals must be equal

2 components of strain:

thermal strain:  $\epsilon_T = \alpha_T \cdot \Delta T$

mechanical ("elastic") strain:

due to forces:  $\epsilon_F = P / EA$

due to curvature:  $\epsilon_\kappa = \kappa \cdot thk_i / 2 = thk_i / (2r)$

```
> KIN1 := alpha_T_1*Delta_T + P_1 / (E_1*thk_1*width_1) + thk_1 /  
(2*r) = alpha_T_2*Delta_T - P_2 / (E_2*thk_2*width_2) - thk_2 /  
(2*r);
```

$$KIN1 := \alpha_{T\_1} \Delta_T + \frac{P}{E\_1 \text{thk\_1 width\_1}} + \frac{\text{thk\_1}}{2r} = \alpha_{T\_2} \Delta_T - \frac{P}{E\_2 \text{thk\_2 width\_2}} - \frac{\text{thk\_2}}{2r} \quad (4)$$

Combining KIN and (EQU,MAT)

> **P:=solve(EQU,MAT,P);**  
**EQU,MATKIN := KIN1;**

$$P := \frac{2(E\_1 I\_1 + E\_2 I\_2)}{r h}$$

$$EQU,MATKIN := \alpha_{T\_1} \Delta_T + \frac{2(E\_1 I\_1 + E\_2 I\_2)}{r h E\_1 \text{thk\_1 width\_1}} + \frac{\text{thk\_1}}{2r} \quad (5)$$

$$= \alpha_{T\_2} \Delta_T - \frac{2(E\_1 I\_1 + E\_2 I\_2)}{r h E\_2 \text{thk\_2 width\_2}} - \frac{\text{thk\_2}}{2r}$$

Second moment of inertia

> **I\_1:=width\_1\*thk\_1^3/12;**  
**I\_2:=width\_2\*thk\_2^3/12;**

$$I_1 := \frac{\text{width\_1 thk\_1}^3}{12}$$

$$I_2 := \frac{\text{width\_2 thk\_2}^3}{12} \quad (6)$$

Resulting equation for curvature radius r

> **r:=simplify(solve(EQU,MATKIN,r));**

$$r := \left( E_1 I_1^2 \text{thk\_1}^4 \text{width\_1}^2 + 3 \text{thk\_2} E_2 \left( \frac{\text{thk\_1}^2}{3} + h \text{thk\_1} + \text{thk\_2} \left( h + \frac{\text{thk\_2}}{3} \right) \right) E_1 \text{width\_2 thk\_1 width\_1} + E_2^2 \text{thk\_2}^4 \text{width\_2}^2 \right) / \quad (7)$$

$$(6 \Delta_T E_1 E_2 h \text{thk\_1 thk\_2 width\_1 width\_2} (-\alpha_{T\_1} + \alpha_{T\_2}))$$

Ratio of Young's moduli n=E\_1/E\_2

> **E\_1:=n\*E\_2;**  
**r:=simplify(eval(r));**

$$E_1 := n E_2 \quad (8)$$

Ratio of thicknesses m=thk\_1/thk\_2 and total height h=thk\_1+thk\_2

> **thk\_1:=m\*h/(1+m);**  
**thk\_2:= h/(1+m);**

$$\text{thk\_1} := \frac{m h}{1 + m}$$

$$\text{thk\_2} := \frac{h}{1 + m} \quad (9)$$

Ratio of widths q=width\_1/width\_2

> **width\_1:=q\*width\_2;**

$$\text{width\_1} := q \text{width\_2} \quad (10)$$

Resulting curvature:

```
> curv:=simplify(1/r);
```

$$curv := - \frac{6 (\alpha_{T\_1} - \alpha_{T\_2}) q m n \Delta T (1 + m)^2}{h (n^2 m^4 q^2 + 4 m^3 n q + 6 m^2 n q + 4 n m q + 1)} \quad (11)$$

Timoshenko's original formula (without variable widths)

```
> curvTimo:=6*(alpha_T_2-alpha_T_1)*Delta_T*(1+m)^2/(h*(3*(1+m)^2+(1+m*n)*(m^2+1/(m*n))));
```

$$curvTimo := \frac{6 (-\alpha_{T\_1} + \alpha_{T\_2}) \Delta T (1 + m)^2}{h \left( 3 (1 + m)^2 + (m n + 1) \left( m^2 + \frac{1}{m n} \right) \right)} \quad (12)$$

## Data Input

Bilayer experiment B5 wet-to-dry

for strain evaluation, a negative sign of  $\alpha_{T\_2}$  is important

Measured values

```
> Delta_T:=1;
   alpha_T_1:=0.0;

   width_1:=9.10;
   width_2:=20.36;
   q:=width_1/width_2;

   thk_1:=1.64;
   thk_2:=1.36;
   m:=thk_1/thk_2;
   h:=thk_1+thk_2;

   n:=299.705696811487; #from optimization
   alpha_T_2:=-0.0509093164628522; #from optimization
```

$$\begin{aligned} \Delta T &:= 1 \\ \alpha_{T\_1} &:= 0. \\ width\_1 &:= 9.10 \\ width\_2 &:= 20.36 \\ q &:= 0.4469548134 \\ thk\_1 &:= 1.64 \\ thk\_2 &:= 1.36 \\ m &:= 1.205882353 \\ h &:= 3.00 \\ n &:= 299.705696811487 \\ \alpha_{T\_2} &:= -0.0509093164628522 \end{aligned} \quad (13)$$

```
> curv;
```

$$-0.001966403201 \quad (14)$$

Evaluation of strains

Thermal strain:  $\epsilon_{s\_T} = \alpha_{T\_i} * \Delta T$  (constant over thickness)

```
> eps_Temp_1 := alpha_T_1*Delta_T;  
eps_Temp_2 := alpha_T_2*Delta_T;  
eps_Temp_1 := 0.  
eps_Temp_2 := -0.0509093164628522
```

(15)

Mechanical/elastic strain

due to normal forces:  $\epsilon_{s\_F} = P / EA$  with  $A = thk\_i * width$  (constant over thickness)

due to curvature:  $\epsilon_{s\_kappa} = kappa * z\_i = z\_i / r$  (linear over thickness, with coordinate  $z\_i$  starting from middle of each layer downwards)

```
> eps_Mech_1F := P_1 / (E_1*thk_1*width_1);  
eps_Mech_1M_top := (-thk_1/2)/r;  
eps_Mech_1M_mid := 0;  
eps_Mech_1M_bottom := (thk_1/2)/r;  
eps_Mech_2F := - P_2 / (E_2*thk_2*width_2);  
eps_Mech_2M_top := (-thk_2/2)/r;  
eps_Mech_2M_mid := 0;  
eps_Mech_2M_bottom := (thk_2/2)/r;  
eps_Mech_1F := -0.0002950752123  
eps_Mech_1M_top := 0.001612450626  
eps_Mech_1M_mid := 0  
eps_Mech_1M_bottom := -0.001612450626  
eps_Mech_2F := 0.04766463643  
eps_Mech_2M_top := 0.001337154177  
eps_Mech_2M_mid := 0  
eps_Mech_2M_bottom := -0.001337154177
```

(16)

Check: total strain on bearing surface (bottom passive and top active) must be equal

```
> eps_Total_1 := eps_Temp_1 + eps_Mech_1F + eps_Mech_1M_bottom;  
eps_Total_2 := eps_Temp_2 + eps_Mech_2F + eps_Mech_2M_top;  
eps_Total_1 := -0.001907525838  
eps_Total_2 := -0.001907525853
```

(17)

Plot strains over thickness

Functions over thickness, parameter  $z$  starting from bottom upwards

```
> eps_Temp_fct:=piecewise(0<=z and z<thk_2,alpha_T_2*Delta_T,  
thk_2<=z and z<=(thk_2+thk_1),alpha_T_1*Delta_T);  
eps_Mech_F_fct:=piecewise(0<=z and z<thk_2,-P_2/(E_2*thk_2*  
width_2),thk_2<=z and z<=(thk_2+thk_1),P_1/(E_1*thk_1*width_1));  
eps_Mech_M_fct:=piecewise(0<=z and z<thk_2,-(-thk_2/2+z)/r,  
thk_2<=z and z<=(thk_2+thk_1),-(-thk_2-thk_1/2+z)/r);
```

$$\begin{aligned} \text{eps\_Temp\_fct} &:= \begin{cases} -0.0509093164628522 & 0 \leq z < 1.36 \\ 0. & 1.36 \leq z \leq 3.00 \end{cases} \\ \text{eps\_Mech\_F\_fct} &:= \begin{cases} 0.04766463643 & 0 \leq z < 1.36 \\ -0.0002950752123 & 1.36 \leq z \leq 3.00 \end{cases} \end{aligned}$$

$$\text{eps\_Mech\_M\_fct} := \begin{cases} -0.001337154177 + 0.001966403202 z & 0 \leq z < 1.36 \\ -0.004286758980 + 0.001966403202 z & 1.36 \leq z \leq 3.00 \end{cases} \quad (18)$$

**> eps\_Total\_fct:=eps\_Temp\_fct+eps\_Mech\_F\_fct+eps\_Mech\_M\_fct;**

$$\begin{aligned} \text{eps\_Total\_fct} &:= \left( \begin{cases} -0.0509093164628522 & 0 \leq z < 1.36 \\ 0. & 1.36 \leq z \leq 3.00 \end{cases} \right) \\ &+ \left( \begin{cases} 0.04766463643 & 0 \leq z < 1.36 \\ -0.0002950752123 & 1.36 \leq z \leq 3.00 \end{cases} \right) \\ &+ \left( \begin{cases} -0.001337154177 + 0.001966403202 z & 0 \leq z < 1.36 \\ -0.004286758980 + 0.001966403202 z & 1.36 \leq z \leq 3.00 \end{cases} \right) \end{aligned} \quad (19)$$

Total strains on top and bottom

$$\begin{aligned} &\textbf{> eps_Tot_top:=eval(eps_Total_fct,z=h);} \\ &\textbf{eps_Tot_bottom:=eval(eps_Total_fct,z=0);} \\ &\text{eps\_Tot\_top} := 0.001317375414 \\ &\text{eps\_Tot\_bottom} := -0.004581834207 \end{aligned} \quad (20)$$

temperatur strain (red)

**> plot\_T:=plot([eps\_Temp\_fct,z,z=0..(thk\_1+thk\_2)],color=red, legend=['eps\_temp']):**

normal strain (blue)

**> plot\_F:=plot([eps\_Mech\_F\_fct,z,z=0..(thk\_1+thk\_2)],color=blue, linestyle=dash,legend=['eps\_normal']):**

bending strain (green)

**> plot\_M:=plot([eps\_Mech\_M\_fct,z,z=0..(thk\_1+thk\_2)],color=green, linestyle=dash,legend=['eps\_bending']):**

elastic strain=normal+bending (purple)

**> plot\_elast:=plot([eps\_Mech\_F\_fct+eps\_Mech\_M\_fct,z,z=0..(thk\_1+thk\_2)],color=purple,legend=['eps\_elastic']):**

total strain (black)

**> plot\_Total:=plot([eps\_Total\_fct,z,z=0..(thk\_1+thk\_2)],color=black,linestyle=dashdot,legend=['eps\_total']):**

display total, elastic and 3 components in one

red: temperature

blue: normal force


red: bending

purple: elastic

black: total

**> display(plot\_elast,plot\_F,plot\_M,plot\_T,plot\_Total,labels=**

```
['strain','z']);
```



>>> high elastic strains of about 5% in the active layer, low total curvature